Approssimazione della soluzione di un sistema lineare

col metodo SOR

Enrico Bertolazzi

Carica le librerie

```
> initialize ;
  with(LinearAlgebra):
  with(plots):
```

initialize

Definisce la procedura Normi

```
> Normi := proc(v)
    local i ;
    return max(seq(abs(v[i]),i=1..Dimension(v))) ;
  end proc :
```

Definisce la procedura SOR

```
> SOR := proc(A, b, xs, n, epsi, omega)
    local i, j, k, resj, x, dm, bf, rlist ;
    rlist := [] ;
    dm    := Dimension(b) ;
    x     := xs ;
    for k from 1 to n do
      bf := 0 ;
      for i from 1 to dm do
        resj := b[i] - add(evalf(A[i,j]*x[j]),j=1..dm) ;
        x[i] := evalf(x[i] + omega * resj/A[i,i]) ;
        if bf < abs(resj) then bf := abs(resj) ; end if ;
      end do ;
      rlist := [ op(rlist), [k,bf] ] ;
      if bf < epsi then break end if ;
    end do ;
    return rlist, x ;
  end proc :
```

Esempio d'uso (matrice simmetrica)

```
> # definisce la matrice
```

```
A := <<2,0,-1,-1>|<0,2,0,-1>|<-1,0,1,0>|<-1,-1,0,2>> ;
```

$$A := \begin{bmatrix} 2 & 0 & -1 & -1 \\ 0 & 2 & 0 & -1 \\ -1 & 0 & 1 & 0 \\ -1 & -1 & 0 & 2 \end{bmatrix}$$

```
> # definisce il termine noto per avere soluzione <1,2,3,4>
  a := A.<1,2,3,4> ;
```

$$a := \begin{bmatrix} -5 \\ 0 \\ 2 \\ 5 \end{bmatrix}$$

```
> # approssima la soluzione con SOR vari omega
  r1, s1 := SOR(A, a, <0,3,2,1>, 15, 1e-6,1) :
  r2, s2 := SOR(A, a, <0,3,2,1>, 15, 1e-6,1.25) :
  r3, s3 := SOR(A, a, <0,3,2,1>, 15, 1e-6,1.5) :
```

```
> # stampa la soluzione
  s1,s2,s3 ;
```

$$\begin{bmatrix} 0.7562298775 \\ 1.899027109 \\ 2.756229877 \\ 3.827628493 \end{bmatrix}, \begin{bmatrix} 0.9469030237 \\ 1.978006510 \\ 2.950215171 \\ 3.964796806 \end{bmatrix}, \begin{bmatrix} 1.000115626 \\ 2.000007564 \\ 2.999958285 \\ 4.000042508 \end{bmatrix}$$

```
> # disegna la norma del residuo
  plot([r1,r2,r3],style=line,thickness=2,color=[red,green,blue],
       labels=["iterate","residuo"]);
```



Esempio d'uso (matrice simmetrica)

```
> # definisce la matrice
  B := <<2,0,-2,1>|<0,2,0,-1>|<-1,0,1,0>|<-3,-1,0,1>> ;
```

$$B := \begin{bmatrix} 2 & 0 & -1 & -3 \\ 0 & 2 & 0 & -1 \\ -2 & 0 & 1 & 0 \\ 1 & -1 & 0 & 1 \end{bmatrix}$$

```
> # definisce il termine noto per avere soluzione <1,2,3,4>
  b := B.<1,2,3,4> ;
```

$$b := \begin{bmatrix} -13 \\ 0 \\ 1 \\ 3 \end{bmatrix}$$

```
> # approssima la soluzione con SOR vari omega
  r1, s1 := SOR(B, b, <0,3,2,1>, 15, 1e-6,0.85) :
  r2, s2 := SOR(B, b, <0,3,2,1>, 15, 1e-6,1.0) :
  r3, s3 := SOR(B, b, <0,3,2,1>, 15, 1e-6,1.1) :
```

```
> # stampa la soluzione
  s1,s2,s3 ;
```

$$\begin{bmatrix} 1.004189023 \\ 2.002056575 \\ 3.006955290 \\ 3.998715642 \end{bmatrix}, \begin{bmatrix} 1.039062500 \\ 2.011718750 \\ 3.078125000 \\ 3.972656250 \end{bmatrix}, \begin{bmatrix} -0.019390788 \\ 1.671780394 \\ 0.734743333 \\ 4.828220521 \end{bmatrix}$$

```
> # disegna la norma del residuo
  plot([r1,r2,r3],style=line,thickness=2,color=[red,green,blue],
       labels=["iterate","residuo"]);
```




```
>
```